

## 第2回 定期発注方式

高桑 宗右エ門, 三輪 冠奈

### 1. はじめに

本シリーズでは、在庫管理の具体的な場面を取り上げ、PCを用いて、それぞれの状況に合った在庫管理方式を適用するために、在庫管理方式およびモデル化の考え方を具体的に解説している。そして、在庫管理方式をいっそう理解しやすくするための手段を提示し、併せて、実際の問題へ応用するための参考事例を紹介することを目的とする。今回は、定期発注方式について述べることにする。

今回においても、前回で述べた方法を踏襲し、初めにExcelを用いて、シミュレーション・モデルを構築する。次に、グラフ上に時々刻々変化の様子を動的に表現するために、構築済みのExcelモデルにアニメーション・モデルを付加するための手順について解説する。ここでも、Excelと併用するシステムとして、Arenaシミュレーション・アニメーション・システムを用いてモデル構築を進めることにする<sup>1</sup>。

### 2. 定期発注方式

定期発注方式 (Periodic Ordering System, Replenishment System) においては、発注間隔が一定で、在庫調査時点で必ず発注が行われる。定期発注方式の在庫モデルを図1に示す。発注間隔は一定であるが、発注量はそのつど変化する。定期的な発注間隔を「発注サイクル」ともいう。発注間隔は、通常、発注側や

たかくわ そうえもん, みわ かな  
名古屋大学 大学院経済学研究科  
〒464-8601 名古屋市千種区不老町

<sup>1</sup> Microsoft Excelは米国Microsoft Corporationの登録商標であり、Arenaは米国Rockwell Software Corporationの登録商標である。本稿以降で取り上げるプログラムを実行するには、これら二つのソフトウェアがインストール済みであることが必要である。Arenaソフトウェアは文献[5]の付録に添付されており、インストールに特別の制約はない。

受注側などの物理的諸条件などにより決定される。次節以降の式の導出や本方式の詳細については、文献[1~4]を参照してほしい。

#### 2.1 記号

本節 (Excelモデル) で用いる記号を以下にまとめて示す。

- $ad$ : 単位期間当たりの需要量の平均
- $AII$ : 平均在庫量
- $ASQ$ : 平均品切れ量
- $C$ : 発注費 (1回あたり)
- $D$ : 年間需要量
- $d$ : 需要量 (1期)
- $ek$ : 実行期間
- $I$ : 在庫量
- $i$ : 年間在庫保管費率
- $k$ : 期
- $L$ : 納入リードタイム
- $OC$ : 発注回数
- $P$ : 製造原価 (1個あたり)
- $POSDATA$ : セルの名称




図1 定期発注方式の在庫モデル

- Q：発注量
- QR：発注残
- SA：安全在庫量
- SC：品切れ回数
- sd：単位期間当たりの需要量の標準偏差
- ST：在庫補充水準
- T：発注間隔
- $\theta$ ：需要へのサービス率を表す信頼係数

## 2.2 計算式

### (1) 安全在庫量

安全在庫量は、発注間隔に納入リードタイムを加えた期間に対する需要量のバラツキに、需要へのサービス率を考慮して、次式で与えられる。

$$SA = \theta \times \sqrt{(L+T)} \times sd \quad (1)$$

### (2) 在庫補充水準

在庫補充水準は、発注間隔に納入リードタイムを加えた期間に対する需要量の期待値と、安全在庫量の和として、次式で与えられる。

$$ST = (L+T) \times ad + SA \quad (2)$$

### (3) 発注量

発注量は、在庫補充水準から、現在の手持ち在庫量および発注残（未入荷分の量）を差し引いた量として、次式で与えられる。

$$Q_k = ST - I_k - QR_k \quad (3)$$

ここに、添え字  $k$  は期を表す。

## 2.3 事例

ある日用品メーカーでは、42種類の製品（洗剤、漂白剤、シャンプーなど）を生産して販売している。いま、原材料メーカー・メーカー（工場）・流通センター（営業所兼務）・小売店・顧客というサプライチェーンに

において、ある営業所における一製品品目の在庫管理について考える。

この営業所では、複数の製品を取り扱っており、製品によって所管の小売店で販売される総売上高に差異があるので、図2に示すような製品1ヶ月分の需要量に関するパレート図を描いてみる。これにより、重点的に管理すべき製品とそうでない製品を区分けすることができ、各製品品目に対して、適切な在庫管理の方法がそれぞれとられることになる。

ここでは、ABC分析により、A品目（重点管理品目）に属する製品番号6を取り上げる。この製品の1ヶ月分の需要データについて、適合度検定を行ったところ、1日の需要は、平均が122、標準偏差が68.6（単位：個）の正規分布に従うことがわかった。また、工場との輸送に関する取り決めにより、発注間隔は5日、納入リードタイムは1日であるので、定期発注方式により在庫管理を実施する様子をシミュレートしてみる。

## 3. シミュレーション・モデル

前回の定量発注方式による場合と同様に、今回の在庫管理方式のシミュレーション・アニメーション・モデルに関する手順は、次の2ステップで構成される。

- (1) Excelによるシミュレーション実行。
- (2) ArenaによるExcelのシミュレーション結果の読み込みと、在庫推移を表示するアニメーションの実行。

### 3.1 Excelによるシミュレーション・モデル

#### (1) データ入力

図3に示した画面において、指定したセルに対応す




図2 製品1ヶ月分の総売上高に関するパレート図




図3 Excelによるシミュレーション設定画面

る値を入力すると、計算式の入力された理論値のセルにその結果が瞬時に表示される。

(2) プログラムの実行

指定した実行期間のシミュレーションを開始するために、Go ボタンをクリックする。すると、ユーザーフォームが表示されるので、実行期間を入力する。

(3) 結果表示

シミュレーション結果が表示される。ロジックのフローチャートを図4に示し、VBAプログラムの詳細を図5に示す。

3.2 セルの内容

(1) データ入力の Excel シート

図3中のセル (D6: D15) に示した項目に対応するデータを入力する。以下に示す計算式に基づいて、セル (D19: D22) に結果が表示される。

安全在庫量(SA) :  $D19 = D14 * \text{SQRT}(D12 + D13) * D10$

在庫補充水準(ST) :  $D20 = D11 * (D12 + D13) + D19$

年間発注回数(n) :  $D21 = 365 / D13$

年間総変動費用(U) :  $D22 = \text{SQRT}(2 * D6 * D7 * D8 * D9)$

(2) シミュレーション結果

シミュレーション実行後の結果がプログラムの実行により、セル (C31: C34) に表示される。その結果




図4 ロジックのフローチャート

```

Private Sub 定期発注方式実行_Click()
'変数定義
Dim ek, Q, k, L, T, ST, I, d, ad, sd, OC, SC, _
    AII, ASQ As Integer
Dim POSDATA As String
'ユーザーフォームの表示
UserForm1.Show
ek = UserForm1.Period.Value      '実行期間設定
'初期値設定
OC = 0
SC = 0
AII = 0
ASQ = 0
sd = Cells(10, 4).Value
ad = Cells(11, 4).Value
L = Cells(12, 4).Value
T = Cells(13, 4).Value
I = Cells(15, 4).Value
ST = Int(Cells(20, 4).Value)
'リードタイム分の表生成
k = 2
Do Until k = L + 2
    Cells(k, 7).Value = k      '期(k)
    Cells(k, 8).Value = 0      '期首手持ち在庫量(qb(k))
    Cells(k, 9).Value = 0      '需要量(d(k))
    Cells(k, 10).Value = 0     '期末手持ち在庫量(qe(k))
    Cells(k, 11).Value = 0     '発注量(pr(k))
    Cells(k, 12).Value = 0     '入庫量(r(k))
    Cells(k, 13).Value = 0     '発注残(qr(k))
    k = k + 1
Loop
'初期在庫量設定
Cells(L + 1, 10).Value = I
'実行期間終了までの各数値の計算
k = L + 2
Do Until k = ek + L + 2
    Cells(k, 7).Value = k - L - 1      '期(k)
    Cells(k, 8).Value = Cells(k - 1, 10).Value + _
        Cells(k - 1, 12).Value
        '期首の手持ち在庫量(qb(k))
    '需要生成
    Randomize      '乱数発生ルーチンを初期化
    d = Application.WorksheetFunction. _
        NormInv(Rnd(), ad, sd)
    If Int(d) > 0 Then
        Cells(k, 9).Value = Int(d)      '需要(d(k)=d(k))
    Else
        Cells(k, 9).Value = 0          '需要(d(k)=0)
    End If
    Cells(k, 10).Value = Cells(k, 8).Value - _
        Cells(k, 9).Value      '期末の手持ち在庫量(qe(k))
    '発注日のとき
    If Int((k - L - 1) / T) = (k - L - 1) / T Then
        Q = ST - Cells(k, 10).Value - Cells(k - 1, 13).Value
        '発注量の計算(Q=ST-qe(k)-qr(k-1))
        If Q > 0 Then      '発注があるとき
            Cells(k, 11).Value = Q      '発注量(pr(k)=Q)
        Else
            Cells(k, 11).Value = 0      '発注量(pr(k)=0)
        End If
    '発注日でないとき
    Else
        Cells(k, 11).Value = 0          '発注量(pr(k)=0)
    End If
    Cells(k, 12).Value = Cells(k - L, 11).Value
        '期末の入庫量(r(k)=pr(k-L))
    Cells(k, 13).Value = Cells(k - 1, 13).Value + _
        Cells(k, 11).Value - Cells(k, 12).Value
        '発注残(qr(k)=qr(k-1)+pr(k)-r(k))

'期平均在庫量の計算(A1)
If Cells(k, 10).Value > 0 Then
    Cells(k, 14).Value = ((Cells(k, 8).Value + _
        Cells(k, 10).Value) / 2)
Else
    If Cells(k, 8).Value > 0 Then
        Cells(k, 14).Value = (Cells(k, 8).Value + _
            Cells(k, 8).Value) / (2 * Cells(k, 9))
    Else
        Cells(k, 14).Value = 0
    End If
End If
AII = AII + Cells(k, 14).Value

'期平均品切れ量の計算(A2)
If Cells(k, 10).Value > 0 Then
    Cells(k, 15).Value = 0
Else
    If Cells(k, 8).Value > 0 Then
        Cells(k, 15).Value = (Cells(k, 10).Value + _
            Cells(k, 10).Value) / (2 * Cells(k, 9))
    Else
        Cells(k, 15).Value = -((Cells(k, 8).Value + _
            Cells(k, 10).Value) / 2)
    End If
End If
ASQ = ASQ + Cells(k, 15).Value

'期発注回数の計算(A3)
If Cells(k, 11).Value > 0 Then
    Cells(k, 16).Value = 1
    OC = OC + 1
Else
    Cells(k, 16).Value = 0
End If

'期品切れ回数の計算(A4)
If Cells(k, 10).Value > 0 Then
    Cells(k, 17).Value = 0
Else
    Cells(k, 17).Value = 1
    SC = SC + 1
End If

k = k + 1
Loop

'最初に生成した表の削除
k = 1
Do Until k = L + 1
    ActiveSheet.Range("G2:R2").Delete (xlShiftUp)
    k = k + 1
Loop
'結果の書き出し
Cells(31, 3).Value = AII / ek      '平均在庫量
Cells(32, 3).Value = ASQ / ek      '平均品切れ量
Cells(33, 3).Value = OC            '発注回数
Cells(34, 3).Value = SC            '品切れ回数
'結果のセルへの名前定義
Worksheets("Sheet1").Range(Worksheets("Sheet1"). _
    Range("G2"), Worksheets("Sheet1").Range("G2"). _
    End(xlToRight).End(xlDown)).Name = "POSDATA"
End Sub

```

図5 VBAプログラムの詳細

から、セル (C 35: C 37) は、以下の計算式に基づいて、計算される。

$$\text{発注費用} : C 35 = C 33 * D 8$$

$$\text{保管費用} : C 36 = C 31 * D 6 * D 7$$

$$\text{総変動費用} : C 37 = C 35 + C 36$$

ここでは、定期発注方式において、年間需要量は 44530 (=122×365) 個、1日の需要の標準偏差は 68.6 個、納入リードタイムは 1日、発注間隔は 5日、サービス率 95%を表す信頼係数は 1.65 として、それぞれの数値を入力する。すると、安全在庫量は 277.3 個、そして在庫補充水準は 1009.3 個がそれぞれ得られる。ここでは、初期在庫量は 1000 個として、365 日分について、シミュレーションの実行の様子を図 6 に示す。図中、第 5 日 (期) において、期末手持ち在庫量は 464 個であるので、発注量は 545 (=1009-464-0) 個となる。この分は、翌日 (第 6 日) に納入される。以降、第 10 日、15 日、20 日、…、に 5 日の間隔で発注される。さらに、前回同様、期平均在庫量、期平均品切れ量、発注回数、品切れ回数 (日数) についても、統計量を表示するようにしている。

#### 4. アニメーション・モデル

前回述べた手順を踏襲して、はじめに Excel を用いて、シミュレーション実験を行う。在庫状況の推移について計数的に把握することができる。次に、在庫量が変化の様子を時系列的にアニメーション表示することを試みる。今回においても、Arena モデルを追加することにより、Excel モデルで実行されたシミュレーション結果を、直接ダイナミックにプロット表示する方法について述べることにする。

##### 4.1 記号

本項 (Arena モデル) で用いる記号を以下にまとめて示す。なお、節 2.1 で既出の記号については省略する。

A1: 期平均在庫量

A2: 期平均品切れ量

A3: 総発注回数

A4: 総品切れ回数

dk: 需要量 (kk 期)

line: 0 を表示するためのプロット変数

kk: 期<sup>2</sup>

期	期首手持ち在庫量	需要	期末手持ち在庫量	発注量	納入量	発注額	期平均在庫量	期平均品切れ量	発注回数	品切れ回数
1	1000	208	792	0	0	0	896	0	0	0
2	792	6	786	0	0	0	789	0	0	0
3	786	96	690	0	0	0	738	0	0	0
4	690	82	608	0	0	0	649	0	0	0
5	608	144	464	545	0	545	536	0	1	0
6	464	121	343	0	545	0	403.5	0	0	0
7	888	0	888	0	0	0	888	0	0	0
8	888	0	888	0	0	0	888	0	0	0
9	888	257	631	0	0	0	759.5	0	0	0
10	631	72	559	450	0	450	595	0	1	0
11	559	176	383	0	450	0	471	0	0	0
12	833	177	656	0	0	0	744.5	0	0	0
13	656	99	557	0	0	0	606.5	0	0	0
14	557	18	539	0	0	0	548	0	0	0
15	539	35	504	505	0	505	521.5	0	1	0
16	504	213	291	0	505	0	397.5	0	0	0
17	796	188	608	0	0	0	702	0	0	0
18	608	158	450	0	0	0	529	0	0	0
19	450	157	293	0	0	0	371.5	0	0	0
20	293	134	159	850	0	850	226	0	1	0
21	159	92	67	0	850	0	113	0	0	0
22	917	51	866	0	0	0	891.5	0	0	0
23	866	145	721	0	0	0	793.5	0	0	0
24	721	8	713	0	0	0	717	0	0	0
25	713	154	559	450	0	450	636	0	1	0
26	559	108	451	0	450	0	505	0	0	0
27	901	43	858	0	0	0	879.5	0	0	0
28	858	94	764	0	0	0	811	0	0	0
29	764	138	626	0	0	0	695	0	0	0
30	626	256	370	639	0	639	498	0	1	0
31	370	157	213	0	639	0	291.5	0	0	0
32	852	116	736	0	0	0	794	0	0	0
33	736	16	720	0	0	0	728	0	0	0
34	720	141	579	0	0	0	649.5	0	0	0
35	579	184	395	614	0	614	487	0	1	0
36	395	39	356	0	614	0	375.5	0	0	0
37	970	94	876	0	0	0	923	0	0	0

図 6 Excel によるシミュレーション結果

<sup>2</sup> Arena では、期を表す変数は *kk* を用いる。これは、節 2.1 の *k* に対応している。

prk : 発注量 (kk 期)  
 qbk : 期首の手持ち在庫量  
 qek : 期末の手持ち在庫量  
 qrk : 発注残  
 qq : 在庫量のプロット変数  
 rk : 納入量

#### 4.2 モデルの実行手順

以下の手順で、アニメーションを実行する。

- [Step1] エンティティを生成する。1期(日)に1個のエンティティが生成される。
- [Step2] 実行された Excel ファイルを読み込む。
- [Step3] プロットに表示する在庫量 (qq) と 0 を表示するプロット変数 (line) を割り付ける。
- [Step4] 発注回数と品切れ回数をカウントする。
- [Step5] エンティティを破棄する。


モデルの詳細を図7にまとめて示す。図中、上側のブロック線図はプログラムのモジュール表示である。

#### 4.3 モジュールの詳細

前回においては、前項[Step2]のファイルの読み込みと、[Step3]のプロット表示について、前稿4.3(1)および(2)節で詳しく述べたので、今回は、前回の設定と異なる箇所のみ言及することにする。詳細については、前稿を参照してほしい。なお、ここでは、アニメーションに追加する変数 (Variable) について特に詳しく述べることにする。

(1) 外部ファイルからの読み込みと書出し

Arenaでは、ReadWriteモジュールによって、外部ソースからファイルを読み込んだり書き込んだりすることができる。今回は、Excelシミュレーション・モデルの実行によって作成されたデータのファイル



モデルのモジュール入力内容

##### [Step1] エンティティの生成

Create モジュール

Name	Create Entity
Time Between Arrivals	Constant
Type	24
Value	Hours
Units	

##### [Step2] データの読み込み

ReadWrite モジュール

Name	Read File of Data
Type	Read from File
Arena File Name	Data File

Assignments		
Type	Variable	
Variable Name	kk	

	Type	Variable Name
1	Variable	kk
2	Variable	qbk
3	Variable	dk
4	Variable	qek
5	Variable	prk
6	Variable	rk
7	Variable	qrk
8	Variable	A1
9	Variable	A2
10	Variable	A3
11	Variable	A4

##### [Step3] 期首在庫量と期末在庫量の割付とプロット変数の設定

(qq=qbk,line=0,qq=qek)

Assign モジュール

Name	Assign qq and variable of plot
Assignments	
Type	Variable
Variable Name	qq
New Value	qbk

	Type	Variable Name	New Value
1	Variable	qq	qbk
2	Variable	line	0
3	Variable	qq	qek

##### [Step4] 発注回数、品切れ回数の増加 (OC=OC+A3,SC=SC+A4)

Assign モジュール

Name	Assign Counter
Assignments	
Type	Variable
Variable Name	OC
New Value	OC+A3

	Type	Variable Name	New Value
1	Variable	OC	OC+A3
2	Variable	SC	SC+A4

##### [Step5] 終了

Dispose モジュール

Name	Dispose Entity
------	----------------

データモジュールの設定内容

File データモジュール

Name	Data File
Access Type	Microsoft Excel (*.xls)
Operating System File Name	POSmodel.xls

Recordsets		
Recordset Name	Recordset 1	
Named Range	POSDATA	

図7 アニメーション・モデルの詳細

(POSmodel.xls)を読み込む。データが読み込まれるセルには、名前を定義しておく必要がある。

本モデルでは、VBAプログラムの実行時に、作成された表のセルに“POSDATA”と名前を定義している。

(2) プロットの設定

プロットの設定は、Animate ツールバーの Plot ボタンを用いて設定する。Expression に、表示する変数 ( $qq, line$ ) を追加する。今回は、Minimum, Maximum には、プロット表示される最小値 -300 と最大値 900 を入力する Time Range には、時間単位を Day として、100 を入力する。

(3) 変数 (Variable) の表示

アニメーション図では、シミュレーションの実行中に、必要に応じて、期 ( $kk$ ) や需要 ( $dk$ ) などの変数を画面上の任意の位置に表示することができる。変数を表示するためには、Animate ツールバーの Variable オプションを用いる。図 8 に示すような要領で、Variable ダイアログの Expression に表示したい変数名を入力する。この図では、変数  $kk$  を 4 桁表示 (\* が 4 個) することを設定している。

アニメーション図に表示する変数名を以下に列挙する。

Period : 期 ( $kk$ )

Demand : 需要量 ( $dk$ )

Inventory on Hand : 手持ち在庫量 ( $qq$ )

No. of Orders : 発注回数 ( $OC$ )

No. of Shortages : 品切れ回数 ( $SC$ )

Order Quantity : 発注量 ( $prk$ )

Arrival Quantity : 納入量 ( $rk$ )

Average Inventory per Period : 期平均在庫量 ( $AI$ )

Average Shortage per Period : 期平均品切れ量 ( $A2$ )

なお、アニメーション図において、Animate ツールバーの Text を用いて、変数に説明のための文字を加えた。

4.4 アニメーションの実行

Excel シミュレーション・モデルの実行後に、上述したモデルを引き続き実行することにより、シミュレーション結果をダイナミックに表示することができる。実行の様子を図 9 に示す。図中、右側の数字等は前節で設定した在庫量などの管理指標の表示である。

5. おわりに

今回は定期発注方式を取り上げて述べた。次回は、2 (ないしダブル・) ビン方式について検討すること




図 8 Variable のダイアログ




図 9 定期発注方式のシミュレーション・アニメーション・モデルの実行の様子

にし、在庫量を表現するための他の方法についても紹介する。

プログラムおよび実行の様子を下記 URL に掲示するので、本稿に関する内容の詳細については参考ないし照会してほしい。

<http://www.stlab.soec.nagoya-u.ac.jp/>

E-mail: [takakuwa@soec.nagoya-u.ac.jp](mailto:takakuwa@soec.nagoya-u.ac.jp)

#### 参考文献

[1] 春日井博：“総合在庫管理システムの設計”，日本経営

出版会，1971.

[2] 水野幸男：“在庫管理入門”，日科技連，1974.

[3] 人見勝人：“新・生産管理工学”，コロナ社，1997.

[4] E. Naddor, *Inventory Systems*, John Wiley & Sons, 1966.

[5] W. D. Kelton, R. P. Sadowski, and D. A. Sadowski : *Simulation with Arena*, 2nd ed., McGraw-Hill, 2001 (高桑宗右エ門監訳：“シミュレーション—Arena を活用した総合的アプローチ” (第2版), コロナ社, 2002).